


ser. 622 (21)
c 212th



DEPARTMENT OF
ENERGY, MINES AND RESOURCES
MINES BRANCH
OTTAWA

*"OFF-LINE TECO" - DATALINE EDITING
WITH TECO COMMANDS ON PAPER TAPE*



D. FRASER

EXTRACTION METALLURGY DIVISION

APRIL 1971

© Crown Copyrights reserved

Available by mail from Information Canada, Ottawa
and at the following Information Canada bookshops

HALIFAX
1735 Barrington Street

MONTREAL
Æterna-Vie Building, 1182 St. Catherine St. West

OTTAWA
171 Slater Street

TORONTO
221 Yonge Street

WINNIPEG
Mall Center Bldg., 499 Portage Avenue

VANCOUVER
657 Granville Street

or through your bookseller

Price 50 cents Catalogue No. M34-20/133

Price subject to change without notice

Information Canada
Ottawa, 1971

Mines Branch Technical Bulletin TB 133

"OFF-LINE TECO" - DATALINE EDITING WITH TECO COMMANDS
ON PAPER TAPE

by

D. Fraser*

- - - - -

ABSTRACT

The TECO EDITOR on the DATALINE TIME-SHARING SYSTEM can be used in a non-conversational mode called Off-Line Teco. Complete file-editing procedures, consisting of conversational Teco commands, can be prepared on punched paper tape for execution by Off-Line Teco. This greatly reduces the cost of program development and modification on the Dataline System, and makes possible lengthy edits that would be too time-consuming with conversational Teco.

This report is a "User's Manual" for Off-Line Teco, and describes how the editing tapes are made and executed. Dataline Teco users will find the Off-Line Teco approach easy to follow and use.

* Scientific Officer, Extraction Metallurgy Division, Mines Branch, Department of Energy, Mines and Resources, Ottawa, Canada.

Direction des Mines

Bulletin technique TB 133

"OFF-LINE TECO" - MISE AU POINT SUR SYSTÈME DATALINE
AVEC COMMANDES TECO PORTÉES SUR BANDE DE PAPIER

par

D. Fraser*

RÉSUMÉ

L'appareil TECO EDITOR du système à utilisation collective DATALINE peut être utilisé dans un mode non-conversationnel appelé "off-line Teco". Les procédures complètes de mise au point des fichiers qui consistent en des commandes conversationnelles Teco peuvent être préparées sur bande de papier perforée en vue de leur exécution par l'Off-Line Teco. Ceci réduit considérablement le coût de l'élaboration et de la mise au point des programmes selon le système Dataline et rend possible de longues corrections qui seraient trop onéreuses avec le mode conversationnel Teco.

Le présent rapport est un "Manuel de l'utilisateur" de "l'Off-Line Teco" et l'auteur y décrit la préparation et l'utilisation des bandes de mise au point. Les utilisateurs du Dataline Teco trouveront l'approche de "l'Off-Line Teco" facile à suivre et à utiliser.

* Agent scientifique, Division de la métallurgie extractive, Direction des Mines, ministère de l'Energie, des Mines et des Ressources, Ottawa, Canada.

CONTENTS

	<u>Page</u>
Abstract	i
Résumé	ii
1. Introduction	1
2. Preparing Off-Line Teco Tapes	2
1) Beginning the Tape	2
2) Typing Teco Commands	2
3) Correcting Typing Errors as the Tape is Made	4
4) Vertical Spacing	6
5) Ending the Tape	7
3. Entering the Tape on the Terminal	8
4. Executing the Commands in 'UPDATE'	8
5. Correcting Command Errors in 'UPDATE'	10
Conclusion	12
Acknowledgement	12
APPENDIX A - Setting Up Off-Line Teco	13
APPENDIX B - Listing of Program XXX	15

1. INTRODUCTION

Lengthy editing with Teco is often necessary when new programs are written and debugged, or when old programs are modified. The Teco editor is normally used in conversational mode; that is, commands are typed and executed in a step-by-step manner on the terminal keyboard. If a DATALINE user must make numerous text insertions and changes in a file, the editing may use an hour or more of terminal time, but only a few seconds of CPU time.

The OFF-LINE TECO system described in this report allows the editing commands used for conversational Teco to be punched on paper tape. In this way, all the Teco commands required for a long and complex editing operation can be prepared off-line using no terminal time. The tape can be read quickly into the DATALINE system, and the commands executed by Teco in a few seconds. This off-line method of using the Teco editor is as convenient as the usual conversational mode, but uses much less terminal time.

Off-Line Teco tapes are easy to prepare. The name of the file to be edited is placed in parentheses at the beginning of the tape. The commands for the editing operations then follow. Extra commands (not part of the terminal version) allow the user to omit lines on the tape if a typing error is made. Vertical spacing can be inserted in Teco printouts for easier reading.

Before Off-Line Teco can work, a copy of the program XXX must be placed on the user's disk area. Information for obtaining a copy of this program and entering it in the Dataline system is given in Appendix A of this report. The purpose of the

program is to translate the paper tape commands into a form usable by Teco, and then cause their execution. Since the program is short, it can be stored permanently on the disk at minimal cost.

The full specifications of Digital Equipment Corporation PDP-10 TECO are given in the PDP-10 Reference Handbook, pages 501 to 523. The same material is available in booklet form from either DEC or the Dataline Company.

2. PREPARING OFF-LINE TECO TAPES

1) Beginning the Tape

After the leader section, the tape must start with the name, in parentheses, of the disk file to be edited. For example, if the file were named TEST.F40, the tape would begin with:

(TEST.F40)

Up to 500 pages of the file will be placed in the Teco buffer, and the pointer set at the beginning. The parentheses are equivalent to:

EBDSK:TEST.F40\$Y500<A>J\$

If the parentheses are omitted or if the file name is incorrect, an error message, "PARENTHESSES?", will be given.

2) Typing Teco Commands

Commands used on the tape are the same as those typed in conversational mode Teco, with the exception that the tape commands may be separated either by one or more dollar signs or by a single altmode. The dollar sign is preferable since the

altmode is a non-printing character.

Suppose that in the previous example the owner of file TEST.F40 wishes to change the statement NUM = 2 to NUMBER = 6. He also wishes to move back two lines and print the next four lines to check that the editing is correct. An Off-Line Teco tape can be typed in several different ways to accomplish this. Three possibilities are:

- (a) (TEST.F40)\$JSNUM\$IBER\$S2\$-D\$I6\$-2L4T\$
- (b) (TEST.F40)SNUM\$4D\$IBER = 6\$-2L4T\$
- (c) (TEST.F40)
SNUM\$4DIBER = 6\$-2L
4T\$

As these examples show, there is considerable flexibility in typing the commands along the lines. So long as the text is acceptable "Teco language", the user may arrange it as he wishes.

There must be a special way of identifying those dollar characters that are being searched, or inserted in a file, from those used as command delimiters. Accordingly, single quotes must be placed around each dollar sign included in the text of the S or I command. Some examples follow:

- | | | |
|----------------------|-----------------|-----------------|
| SA'\$'B\$-D\$I'\$ | search for A\$B | change to A\$' |
| SA'\$''\$-D\$I''\$' | search for A\$' | change to A\$\$ |
| S'\$''''\$-D\$I'A'\$ | search for \$\$ | change to \$'A' |

All advanced Teco commands can be used with Off-Line Teco. However, it is suggested that users requiring these commands consult Appendix B for a summary of the conversion rules which the translating program XXX applies to the paper tape text.

3) Correcting Typing Errors as the Tape is Made

Off-Line Teco has commands which cause omission of one or more lines already punched on the paper tape so that correct lines can be typed in their place. These commands are generally more useful than the Backspace and Rubout since these keys are impractical to use with more than about eight wrong characters. The commands which will omit lines from the paper tape take the form:

`/OMIT/` or `/OMIT-n/` where `n` is an integer from 1 to 9

The rules for these commands are:

- (a) Any text on the line up to the command itself is omitted. If any text (other than CR-LF) follows the right hand slash of the command, it becomes a new line in place of the omitted text.
- (b) When the command is of the form `/OMIT-n/`, `n` preceding lines are deleted in addition to the above action. The number `n` must be in the range from 1 to 9.

Several examples are given below to illustrate the rules. In every case the user was trying to type the slogan

NOW IS THE
TIME FOR ALL
GOOD MEN TO
COME TO THE AID
OF THEIR COUNTRY.

The examples have not been completed, but enough is shown in each one to indicate how the user corrected his tape to read properly. The text which will be ignored is underlined.

NOW IS THE
TIMEFOR/OMIT/
TIME FOR ALL

NOW IS THE
TIMEFOR /OMIT/TIME FOR ALL

GOOD MEN TO

↑ new line may begin immediately after the slash

NOW IS THE
TIMEFOR ALL
GOOD MEN/OMIT-1/
TIME FOR ALL

NOW IS THE
TIMEFOR ALL
GOOD MEN /OMIT-1/TIME FOR ALL

GOOD MEN TO

↑ omit present line and previous one

NOW IS THE
TIME FORALL
GOOD MEN TO
/OMIT-2/
TIME FOR ALL

NOW IS THE
TIMEFOR ALL
/OMIT-1/TIME FOR ALL
GOOD MEN TO
COME TO THE AID

Extension of the above examples to command strings typed on the Teco tape is straightforward. In the following example, the user wished to replace NUM with KLM but erroneously typed -2D instead of -3D. As before, the underlined text will be ignored.

\$\$NUM\$-2D\$IKLM/OMIT/SNUM\$-3D
\$IKLM\$

\$\$NUM\$-2D
IKL/OMIT-1/
\$\$NUM\$-3D\$IKLM\$

\$\$NUM\$
-2D\$
IKLM\$/OMIT-1/-3D
IKLM\$

It is also worth remembering that many errors in Teco editing can be corrected by giving commands further along the Off-Line Teco tape that will "undo" the wrong editing. If the editing mistake is not too serious, it is often faster to do this than to punch another tape. Some examples will indicate the general approach.

```
$SDOG$-3D$ICAT$  
---- etc ----  
J$SCAT$-3D$IFOX$
```

DOG was to be changed to FOX, but it was erroneously written as CAT. Further along the tape, the pointer was initialized (J) and a search was made for CAT, which was then replaced by FOX.

```
$SNUM$I6$  
---- etc ----  
J$SNUM6$-D$S2$  
-D$I6$
```

NUM = 2 was supposed to be changed to NUM = 6, but an editing mistake gave NUM6 = 2. Subsequent commands corrected this.

It should be noted that two successive control G characters ($\uparrow G \uparrow G$) cannot be used on an Off-Line Teco tape to delete wrong lines.

4) Vertical Spacing

Three commands in Off-Line Teco will move the teletype sheet vertically one, two or three lines. It is helpful to use these commands before Type instructions on the paper tape, so that Teco will put spaces between the various lines when they are printed on the terminal.

/SPACE/	move the teletype sheet up one line
/DSPACE/	move sheet up two lines (double-space)
/TSPACE/	move sheet up three lines (triple-space)

Examples:

```
$$SNUM = $OTT$SA = .7$  
/SPACE/$OTT$SB = $  
-2L/DSPACE/4T$
```

Put one blank line between the first and second lines; double-space before printing another four lines (4T). Dollar signs before and after the slashes are optional.

```
J <SNUM = $; /SPACE/OTT>
```

Type every line in the file with the text "NUM =". Each line will be separated on the teletype by one blank line.

```
$$SABC$ID$OTT$  
/TSPACE//SPACE/$HT$
```

Change ABC to ABCD and print that line. Then space four lines and print the buffer contents (HT).

```
$$SXYZ/SPACE/M$  
$IAB/OMIT/CD$
```

Ambiguous. The legitimate text of a search (S) or insert (I) command should not contain characters resembling Off-Line Teco commands which take precedence.

```
$$SXYZ/SPACE↑XM$  
$IAB/OMIT$I/CD$
```

Correct. The text of the S command is "disguised" with ↑X (match arbitrary character) in place of "/". The I command has been broken into two parts.

5) Ending the Tape

In general, there are two ways to end the tape. One method is to put on the tape the same instructions that would be typed on the terminal to exit from conversational Teco. The easiest command to give is EX\$. Alternatively, the tape can simply stop at the final editing command. This technique is probably the more useful since it allows an opportunity for conversational commands to be typed on the terminal after the Off-Line Teco program has finished. This is explained further in section 4, "Executing the Commands in UPDATE".

3. ENTERING THE TAPE ON THE TERMINAL

Use the RTAPE CUSP to read the paper tape into a file called UPDATE. The name of this disk file is fixed and the procedure will not work with other file names.

```
.R RTAPE
FILE NAME?: UPDATE
SEQUENCE?:
PLEASE TURN ON READER NOW.
----- } paper tape is read in
----- }
↑Z
EXIT
```

4. EXECUTING THE COMMANDS IN 'UPDATE'

Call Teco and when the asterisk is given, type the following one line command:

```
.R TECO
* ERXXX$YHXX$MX$$      where $ means altmode
```

Teco will then execute all the commands that were placed in the file named UPDATE. There are three points that should be noted.

- (1) The final command punched on the paper tape will determine what Teco does when the editing sequence is finished. If the tape ended with an EX\$ command (or equivalent), the message "EXIT" will be given and control will pass to the monitor. Otherwise, Teco will print an asterisk to indicate completion and will wait for further commands to be typed on the terminal. At this time, the Teco buffer will contain

the newly edited version of the file. If the user wishes, he may continue editing the file in conversational mode, using altmodes as delimiters instead of dollar signs. When the editing is complete, the user must type an EX\$ command (or equivalent) so that Teco writes out the edited file and exits to the monitor. Either procedure will leave three files on the disk - the newly edited file, a copy of the original file, and the paper tape file UPDATE.

- (2) To prevent loss of the file being edited, Off-Line Teco automatically creates a duplicate of the original file. The name of this file will consist of the edited file name plus the extension ".BAK". If Teco returned to the monitor by an EX\$ command (or equivalent), the original file can be restored by giving the two monitor level commands:

```
.DELETE filnam  
.RENAME filnam = filnam.BAK
```

However, if the Off-Line Teco procedure is stopped with the ↑C command, no backup file is needed since the original will not have been altered.

- (3) Infrequently, an error message may occur when Teco is executing the edit instructions placed in the file UPDATE. The message will be followed by an asterisk and Teco will then stop. To identify the command causing the problem, type a question mark on the terminal. Teco will respond by printing a short sequence of commands from the paper tape input. The last command printed is the one at which Teco

stopped. Since the command problem in UPDATE must be corrected before the editing can proceed, type a ↑C to return to the monitor. Do not give an EX\$ command, or damage may occur to the file which is being edited. An example follows:

```
.R TECO
*ERXXX$YHXX$MX$$
?SEARCH FAILURE
* ?
<A>J$$NUMBER =$?SEARCH FAILURE
↑C
EXIT
```

The following section covers the correction of minor errors in the command file UPDATE.

5. CORRECTING COMMAND ERRORS IN 'UPDATE'

The procedure for obtaining a printout of the command sequence at which Teco stopped was given above. The user should have no difficulty in matching this command sequence with the corresponding commands in UPDATE. Incidentally, the commands printed by Teco in response to the question mark may differ slightly from those in UPDATE by having extra symbols or missing dollar signs. These minor changes should be ignored - they are internally generated by the Off-Line Teco program and are not the source of the error. The command in UPDATE which Teco could not execute should be examined. It may be logically incorrect or contain wrong characters. If no error exists in the last command,

then one of the previous edit commands is in error. Some examples will illustrate these points.

*? \$S+1.6\$?SEARCH FAILURE

A check of the file being edited indicated a blank between the plus sign and the figure "1". The command should have been \$S+ 1.6\$

*? -20L\$\$NUMBER\$?SEARCH FAILURE

The user had gone back (-20L) to edit the line containing the text "NUMBER". A recount of the lines in the file indicated that the pointer should have been moved back about 25 lines.

*? ATTEMPT TO READ COMMANDS BEYOND DOUBLE ALTMODE

Teco would not print any command for this error message. When UPDATE was examined, a type command (OTT) was found using the letter O, instead of zero.

Errors that may occasionally occur in the file UPDATE are best corrected with conversational Teco. This should cause no difficulties, since the text in UPDATE can be edited like any other file. The dollar signs in UPDATE, however, must not be confused with the altmodes that are typed in conversational Teco.

As an example, the second error condition given above can be removed by substituting -30L for -20L so that the search for "NUMBER" begins well ahead of its occurrence.

.TE UPDATE

```

      |-----| altmodes
*S$-20L$$-4C$DI3$$
      |-----| dollar signs (SHIFT-4)

*EX$$
EXIT
```

CONCLUSION

Off-Line Teco allows file modifications to be effected easily and economically. The advantages of paper tape input are twofold. Firstly, the editing procedure can be done without the "pressure" of using valuable terminal time, and individual commands can be thoroughly checked for accuracy and completeness. Any tape errors can be corrected with the /OMIT/ class of commands. Secondly, very long file edits that are costly in the conversational mode can be performed by tape. For example, Off-Line Teco has frequently been used to edit and replace over one hundred program statements with a single tape.

ACKNOWLEDGEMENT

The author wishes to thank Mr. R. F. Pilgrim for many helpful suggestions during the development of the Off-Line Teco program, and Dr. J. E. Dutrizac for assistance in preparing this report.

APPENDIX A - SETTING UP OFF-LINE TECO


Off-Line Teco requires the use of two disk files, called XXX and UPDATE. The user should first ensure that these names do not conflict with those of other files currently stored on the disk area. The file XXX must contain the Off-Line Teco translating program. A paper tape copy of this program is available from:

Mr. F. J. Kelly
Extraction Metallurgy Division
Mines Branch
300 LeBreton Street
Ottawa, Canada
K1A 0G1
Telephone (613) 994 - 5002

Since the Off-Line Teco program occupies only one block, the file XXX can be left permanently on the disk. The paper tape program copy can be entered into XXX with the RTAPE CUSP, as follows:

```
.R RTAPE  
FILE NAME?: XXX  
SEQUENCE?:  
PLEASE TURN ON READER NOW.  
----  
↑Z  
EXIT
```

The other disk file, UPDATE, is a temporary file for holding the editing commands which are punched by the user on paper tape. The use of this file has been covered in the main part of this report. These are the only preparations for using Off-Line Teco.



APPENDIX B - LISTING OF PROGRAM XXX

- PROGRAM -

- COMMENTS -

```

HK$ERUPDATE$Y20<A>$J$<S/OMIT/↵
$;-K$>J<S/OMIT/$;0K$>J$
<S/OMIT-↑X/$;-3C\K$0K$D$1A-13"E$2D'$>J$
<S/SPACE/$;-7DI$↑A$13I10I$1↑A$>J$
<S/DSPACE/$;-8DI$↑A$13I10I10I$1↑A$>J$
<S/TSPACE/$;-8DI$↑A$13I10I10I10I$1↑A$>J$
!START!$1A-39"L$D$OSTART$
'$1A-40"E$.U0$2C$:S)$"N.-Q0-13"L$OGO$''
↑APARENTHESSES?↑A$↑G$
!GO!JS($-D@I#$EBDSK:##$)-D@I#$Y500<A>J$##J$
<S'$;$;-3D127I$>J$
<S$$;!TAG!$1A-36"E$DOTAG$'-D@I##$>J$
!LOOP!$1A-127"E$DI$$'#$C.-Z"N$OLOOP$'$
HX0$HK$M0$
    
```

```

read UPDATE; perform /OMIT/↵ command
perform /OMIT/[text] command
do /OMIT-n/↵ and /OMIT-n/[text] commands
insert coding for /SPACE/ (CR-LF)
insert coding for /DSPACE/ (CR-LF-LF)
insert coding for /TSPACE/ (CR-LF-LF-LF)
delete leading characters less than 478
("?" If so, find ")" in next 13 chars.
error message and monitor return
modify text to give EBDSK:filnam$Y500<A>J$
replace '$' with 177g (rub out character)
reduce dollar strings to single altmode
replace 177g with dollar character ($)
buffer text to register 0; kill buffer;
execute contents of register 0
    
```

NOTE: In this listing, \$ = dollar character (SHIFT-4)
 \$ = altmode (CTRL-SHIFT-K)
 ↑x = CTRL-x, where x is any character

Q-register 0 (zero) is not available in Off-Line Teco